
10 REM COMM.BAS
20 REM Program to communicate with a Love 300 or 1600 series control
30 REM This program has been tested on 286, 386, 486, and P5 processors
40 REM This program has been tested with BASIC, BASICA, GWBASIC, QuickBasic
4.5, and QBASIC

100 CLS
110 CBA% = &H3F8: REM Base address for Com1. Use &H2F8 for Com2
120 OPEN "com1: 9600,N,8,1,cs" FOR RANDOM AS #1: REM open port 1, 9600 baud
130 RTS% = INP(CBA%+4):       REM  ) Set up variables for comm
140 RTSR% = RTS% AND &HFD:    REM  ) direction control
150 RTST% = RTS% OR &H2:      REM  )
160 STX$ = CHR$(2): REM start transmission character
170 ETX$ = CHR$(3): REM end transmission character
180 ACK$ = CHR$(6): REM acknowledge character
190 NUL$ = CHR$(0): REM null character - used to pad data packet to correct
for UARTs that show the transmit buffer empty when the last chatacter is
being sent

200 TXCMD$ = STX$ + "L32" + "00" + "C5" + ETX$ + NUL$
210 REM L32 = control address
220 REM  00 = read command
230 REM  C5 = checksum
300 PRINT TXCMD$; SPC(10); : REM show command on screen
400 GOSUB 1000: REM transmit / receive subroutine
500 PRINT RXDATA$
600 KY$ = INKEY$
610 IF KY$ = "" THEN GOTO 200
700 CLOSE #1
800 END

1000 TMO% = 0: REM Set up time out counter
1010 RXDATA$ = "":  REM Clear receive buffer

2000 OUT CBA%+4, RTST%: REM set RTS line
2010 PRINT #1, TXCMD$; : REM Send data packet.  Don't forget semicolon!
2020 IF (INP(CBA%+5) AND &H60) <> &H60 THEN GOTO 2020
2030 REM wait until xmit buffer is empty

3000 OUT CBA%+4, RTSR%:  REM reset RTS line
3010 IF LOC(1) = 0 THEN GOTO 3050: REM if receive buffer empty skip read
3020 RX$ = INPUT$(1, #1): REM fetch character
3030 RXDATA$ = RXDATA$ + RX$: REM Add to the end
3040 IF INSTR(1, RX$, ACK$) > 0 THEN GOTO 3080: REM End of RX packet?
3050 TMO% = TMO% + 1: REM time out counter
3060 IF TMO% < 2500 THEN GOTO 3010
3070 RXDATA$ = "Time Out"
3080 RETURN


